

15

Economic Equilibria and Pricing

Plus ce change, plus ce la meme chose.
-Alphonse Karr: "Les Guepes", 1849

15.1 What is an Equilibrium?

As East and West Germany were about to be re-united in the early 1990's, there was considerable interest in how various industries in the two regions would fare under the new economic structure. Similar concerns existed about the same time in Canada, the United States, and Mexico, as trade barriers were about to be dropped under the structure of the new North American Free Trade Agreement (NAFTA). Some of the planners concerned with NAFTA used so-called economic equilibrium models to predict the effect of the new structure on various industries. The basic idea of an equilibrium model is to predict what the state of a system will be in the "steady state", under a new set of external conditions. These new conditions are typically things like new tax laws, new trading conditions, or dramatically new technology for producing some product.

Equilibrium models are of interest to at least two kinds of decision makers: people who set taxes, and people who are concerned with appropriate prices to set. Suppose state X feels it would like to put a tax on littering with, say, glass bottles. An explicit tax on littering is difficult to enforce. Alternatively, the state X might feel it could achieve the same effect by putting a tax on bottles when purchased, and then refunding the tax when the bottle is returned for recycling. Both of these are easy to implement and enforce. If a neighboring state, Y , however, does not have a bottle refund, then citizens of the state Y will be motivated to cross the border to X and turn their bottles in for refund. If the refund is high, then the refund from state X may end up subsidizing bottle manufacturing in state Y . Is this the intention of state X ? A comprehensive equilibrium model takes into account all the incentives of the various sectors or players.

If one is modeling an economy composed of two or more individuals, each acting in his or her self-interest, there is no obvious overall objective function that should be maximized. In a market, a solution, or equilibrium point, is a set of prices such that supply equals demand for each commodity. More generally, an equilibrium for a system is a state in which no individual or component in the system is motivated to change the state. Thus, at equilibrium in an economy, there are no arbitrage possibilities (e.g., buy a commodity in one market and sell it in another market at a higher price at no risk). Because

economic equilibrium problems usually involve multiple players, each with their own objective, these problems can also be viewed as multiple criteria problems.

15.2 A Simple Simultaneous Price/Production Decision

A firm that has the choice of setting either price or quantity for its products may wish to set them simultaneously. If the production process can be modeled as a linear program and the demand curves are linear, then the problem of simultaneously setting price and production follows.

A firm produces and sells two products A and B at price P_A and P_B and in quantities X_A and X_B . Profit maximizing values for P_A , P_B , X_A , and X_B are to be determined. The quantities (sold) are related to the prices by the demand curves:

$$\begin{aligned} X_A &= 60 - 21 P_A + 0.1 P_B, \\ X_B &= 50 - 25 P_B + 0.1 P_A. \end{aligned}$$

Notice the two products are mild substitutes. As the price of one is raised, it causes a modest increase in the demand for the other item.

The production side has the following features:

	Product	
	A	B
Variable Cost per Unit	\$0.20	\$0.30
Production Capacity	25	30

Further, the total production is limited by the constraint:

$$X_A + 2X_B \leq 50.$$

The problem can be written in LINGO form as:

```

MIN = -(PA - 0.20) * XA - (PB - 0.30) * XB;
XA + 21 * PA - 0.1 * PB = 60;
! Demand curve definition;
XB + 25 * PB - 0.1 * PA = 50;
XA <= 25;           !Supply restrictions;
XB <= 30;
XA + 2 * XB <= 50;

```

The solution is:

Optimal solution found at step:		4
Objective value:		-51.95106
Variable	Value	Reduced Cost
PA	1.702805	0.0000000
XA	24.39056	0.0000000
PB	1.494622	0.0000000
XB	12.80472	0.0000000
Row	Slack or Surplus	Dual Price
1	-51.95106	1.000000
2	0.0000000	1.163916
3	0.0000000	0.5168446
4	0.6094447	0.2531134E-07
5	17.19528	0.0000000
6	0.0000000	0.3388889

Note it is the joint capacity constraint $X_A + 2X_B \leq 50$, which is binding. The total profit contribution is \$51.951058.

15.3 Representing Supply & Demand Curves in LPs

The use of smooth supply and demand curves has long been a convenient device in economics courses for thinking about how markets operate. In practice, it may be more convenient to think of supply and demand in more discrete terms. What is frequently done in practice is to use a sector approach for representing demand and supply behavior. For example, one represents the demand side as consisting of a large number of sectors with each sector having a fairly simple behavior. The most convenient behavior is to think of each demand sector as being represented by two numbers:

the maximum price (its reservation price) the sector is willing to pay for a good, and
the amount the sector will buy if the price is not above its reservation price.

The U.S. Treasury Department, when examining the impact of proposed taxes, has apparently represented taxpayers by approximately 10,000 sectors, see Glover and Klingman (1977) for example.

The methodology about to be described is similar to that used in the PIES (Project Independence Evaluation System) model developed by the Department of Energy. This model and its later versions were extensively used from 1974 onward to evaluate the effect of various U.S. energy policies.

Consider the following example. There is a producer A and a consumer X who have the following supply and demand schedules for a single commodity (e.g., energy):


Producer A		Consumer X	
Market Price per Unit	Amount Willing To Sell	Market Price per Unit	Amount Willing To Buy
\$1	2	\$9	2
2	4	4.5	4
3	6	3	6
4	8	2.25	8

For example, if the price is less than \$2, but greater than \$1, then the producer will produce 2 units. However, the consumer would like to buy at least 8 units at this price. By inspection, note the equilibrium price is \$3 and any quantity.

It is easy to find an equilibrium in this market by inspection. Nevertheless, it is useful to examine the LP formulation that could be used to find it. Although there is a single market clearing price, it is useful to interpret the supply schedule as if the supplier is willing to sell the first 2 units at \$1, the next 2 units at \$2 each, etc. Similarly, the consumer is willing to pay \$9 each for the first 2 units, \$4.5 for the next 2 units, etc. To find the market-clearing price such that the amount produced equals the amount consumed, we act as if there is a broker who actually buys and sells at these marginal prices, and all transactions must go through the broker. The broker maximizes his profits. The broker will continue to increase the quantity of goods transferred as long as he can sell it at a price higher than his purchase price. At the broker's optimum, the quantity bought equals the quantity sold and the price offered by the buyers equals the price demanded by the sellers. This satisfies the conditions for a market equilibrium.

Graphically, the situation is as in Figure 15.1:

Figure 15.1 Demand and Supply Curves



The area marked “producer-consumer surplus” is the profit obtained by the hypothetical broker. In reality, this profit is allocated between the producer and the consumer according to the equilibrium price. In the case where the equilibrium price is \$3, the consumer's profit or surplus is the portion of the producer-consumer surplus area above the \$3 horizontal line, while the producer's profit or surplus is the portion of the producer-consumer surplus area below \$3.

Readers with a mathematical bent may note the general approach we are using is based on the fact that, for many problems of finding an equilibrium, one can formulate an objective function that, when optimized, produces a solution satisfying the equilibrium conditions.

For purposes of the LP formulation, define:

$A1$ = units sold by producer for \$1 per unit;
 $A2$ = units sold by producer for \$2 per unit;
 $A3$ = units sold by producer for \$3 per unit;
 $A4$ = units sold by producer for \$4 per unit;
 $X1$ = units bought by consumer for \$9 per unit;
 $X2$ = units bought by consumer for \$4.5 per unit;
 $X3$ = units bought by consumer for \$3 per unit;
 $X4$ = units bought by consumer for \$2.25 per unit.

The formulation is:

```

MAX = 9 * X1 + 4.5 * X2 + 3 * X3 + 2.25 * X4
! Maximize broker's revenue;
- A1 - 2 * A2 - 3 * A3 - 4 * A4;
! minus cost;
A1 + A2 + A3 + A4 - X1 - X2 - X3 - X4 = 0;
! Supply = demand;
A1 <= 2;
A2 <= 2;
A3 <= 2;
A4 <= 2;
! Steps in supply;
X1 <= 2;
X2 <= 2;
X3 <= 2;
X4 <= 2;
! and demand functions;
  
```

A solution is:

$$A1 = A2 = A3 = X1 = X2 = X3 = 2$$

$$A4 = X4 = 0$$

Note there is more than one solution, since $A3$ and $X3$ cancel each other when they are equal.

The dual price on the first constraint is \$3. In general, the dual price on the constraint that sets supply equal to demand is the market-clearing price.

Let us complicate the problem by introducing another supplier, B , and another consumer, Y . Their supply and demand curves are, respectively:

Producer B		Consumer Y	
Market Price per Unit	Amount Willing To Sell	Market Price per Unit	Amount Willing To Buy
\$2	2	\$15	2
4	4	8	4
6	6	5	6
8	8	3	8

An additional complication is shipping costs \$1.5 per unit shipped from A to Y , and \$2 per unit shipped from B to X . What will be the clearing price at the shipping door of A , B , X , and Y ? How much will each participant sell or buy?

The corresponding LP can be developed if we define $B1, B2, B3, B4, Y1, Y2, Y3$ and $Y4$ analogous to $A1, X1$, etc. Also, we define AX, AY, BX , and BY as the number of units shipped from A to X , A to Y , B to X , and B to Y , respectively. The formulation is:

```

MAX = 9 * X1 + 4.5 * X2 + 3 * X3 + 2.25 * X4
      + 15 * Y1 + 8 * Y2 + 5 * Y3 + 3 * Y4
      - 2 * BX - 1.5 * AY - A1 - 2 * A2 - 3 * A3
      - 4 * A4 - 2 * B1 - 4 * B2 - 6 * B3 - 8 * B4;
! Maximize revenue - cost for broker;
- AY + A1 + A2 + A3 + A4 - AX = 0;
! amount shipped from A;
- BX + B1 + B2 + B3 + B4 - BY = 0;
! amount shipped from B;
- X1 - X2 - X3 - X4 + BX + AX = 0;
! amount shipped from X;
- Y1 - Y2 - Y3 - Y4 + AY + BY = 0;
! amount shipped from Y;
A1 <= 2;
A2 <= 2;
A3 <= 2;
A4 <= 2;
B1 <= 2;
B2 <= 2;
B3 <= 2;
B4 <= 2;
X1 <= 2;
X2 <= 2;
X3 <= 2;
X4 <= 2;
Y1 <= 2;
Y2 <= 2;
Y3 <= 2;
Y4 <= 2;

```

Notice from the objective function that the broker is charged \$2 per unit shipped from B to X and \$1.5 per unit shipped from A to Y . Most of the constraints are simple upper bound (SUB) constraints. In realistic-size problems, several thousand SUB-type constraints can be tolerated without adversely affecting computational difficulty.

The original solution is:

Optimal solution found at step:		3
Objective value:		21.00000
Variable	Value	Reduced Cost
X1	2.000000	0.000000
X2	2.000000	0.000000
X3	2.000000	0.000000
X4	0.000000	0.750000
A1	2.000000	0.000000
A2	2.000000	0.000000
A3	2.000000	0.000000
A4	0.000000	1.000000
Row	Slack or Surplus	Dual Price
1	21.00000	1.000000
2	0.000000	-3.000000
3	0.000000	2.000000
4	0.000000	1.000000
5	0.000000	0.000000
6	2.000000	0.000000
7	0.000000	6.000000
8	0.000000	1.500000
9	0.000000	0.000000
10	2.000000	0.000000

From the dual prices on rows 2 through 5, we note the prices at the shipping door of A , B , X , and Y are \$3.5, \$5, \$3.5, and \$5, respectively. At these prices, A and B are willing to produce 6 and 4 units, respectively. While, X and Y are willing to buy 4 and 6 units, respectively. A ships 2 units to Y , where the \$1.5 shipping charge causes them to sell for \$5 per unit. A ships 4 units to X , where they sell for \$3.5 per unit. B ships 4 units to Y , where they sell for \$5 per unit.

15.4 Auctions as Economic Equilibria

The concept of a broker who maximizes producer-consumer surplus can also be applied to auctions. LP is useful if features that might be interpreted as bidders with demand curves complicate the auction. The example presented here is based on a design by R. L. Graves for a course registration system used since 1981 at the University of Chicago in which students bid on courses. See Graves, Sankaran, and Schrage (1993).

Suppose there are N types of objects to be sold (e.g., courses) and there are M bidders (e.g., students). Bidder i is willing to pay up to b_{ij} , $b_{ij} \geq 0$ for one unit of object type j . Further, a bidder is interested in at most one unit of each object type. Let S_j be the number of units of object type j available for sale.

There is a variety of ways of holding the auction. Let us suppose it is a sealed-bid auction and we want to find a single, market-clearing price, p_j , for each object type j , such that:

- at most, S_j units of object j are sold;
- any bid for j less than p_j does not buy a unit;
- $p_j = 0$ if less than S_j units of j are sold;
- any bid for j greater than p_j does buy a unit.

It is easy to determine the equilibrium p_j 's by simply sorting the bids and allocating each unit to the higher bidder first. Nevertheless, in order to prepare for more complicated auctions, let us consider how

to solve this problem as an optimization problem. Again, we take the view of a broker who sells at as high a price as possible (buys at as low) and maximizes profits.

Define:

$$x_{ij} = 1 \text{ if bidder } i \text{ buys a unit of object } j, \text{ else } 0.$$

The LP is:

$$\begin{aligned} \text{Maximize} \quad & \sum_{i=1}^M \sum_{j=1}^N x_{ij} b_{ij} \\ \text{subject to} \quad & \sum_{i=1}^M x_{ij} \leq S_j \text{ for } j = 1 \text{ to } N \\ & x_{ij} \leq 1 \text{ for all } i \text{ and } j. \end{aligned}$$

The dual prices on the first N constraints can be used, with minor modification, as the clearing prices p_j . The possible modifications have to do with the fact that, with step function demand and/or supply curves, there is usually a small range of acceptable clearing prices. The LP solution will choose one price in this range, usually at one end of the range. One may wish to choose a price within the interior of the range to break ties.

Now, we complicate this auction slightly by adding the condition that no bidder wants to buy more than 3 units total. Consider the following specific situation:

		Maximum Price Willing To Pay				
		Objects				
		1	2	3	4	5
Bidders	1	9	2	8	6	3
	2	6	7	9	1	5
	3	7	8	6	3	4
	4	5	4	3	2	1
Capacity		1	2	3	3	4

For example, bidder 3 is willing to pay up to 4 for one unit of object 5. There are only 3 units of object 4 available for sale.

We want to find a “market clearing” price for each object and an allocation of units to bidders, so each bidder is willing to accept the units awarded to him at the market-clearing price. We must generalize the previous condition d to d' : a bidder is satisfied with a particular unit if he cannot find another unit with a bigger difference between his maximum offer price and the market clearing price. This is equivalent to saying each bidder maximizes his consumer surplus.

The associated LP is:

```

MAX = 9 * X11 + 2 * X12 + 8 * X13 + 6 * X14
      + 3 * X15 + 6 * X21 + 7 * X22 + 9 * X23
      + X24 + 5 * X25 + 7 * X31 + 8 * X32 + 6 * X33
      + 3 * X34 + 4 * X35 + 5 * X41 + 4 * X42
      + 3 * X43 + 2 * X44 + X45;
      !(Maximize broker revenues);
X11 + X21 + X31 + X41 <= 1;
      !(Units of object 1 available);
X12 + X22 + X32 + X42 <= 2;          !           .;
X13 + X23 + X33 + X43 <= 3;          !           .;
X14 + X24 + X34 + X44 <= 3;          !           .;
X15 + X25 + X35 + X45 <= 4;
      !(Units of object 5 available);
X11 + X12 + X13 + X14 + X15 <= 3;
      !(Upper limit on buyer 1 demand);
X21 + X22 + X23 + X24 + X25 <= 3;    !           .;
X31 + X32 + X33 + X34 + X35 <= 3;    !           .;
X41 + X42 + X43 + X44 + X45 <= 3;
      !(Upper limit on buyer 2 demand);
X11 <= 1;
X21 <= 1;
X31 <= 1;
X41 <= 1;
X12 <= 1;
X22 <= 1;
X32 <= 1;
X42 <= 1;
X13 <= 1;
X23 <= 1;
X33 <= 1;
X43 <= 1;
X14 <= 1;
X24 <= 1;
X34 <= 1;
X15 <= 1;
X25 <= 1;
X35 <= 1;
X45 <= 1;

```

The solution is:

```

Optimal solution found at step:          23
Objective value:                        67.00000
Variable      Value      Reduced Cost
X11           1.000000      0.000000
X12           0.000000      4.000000
X13           1.000000      0.000000
X14           1.000000      0.000000
X15           0.000000      0.000000
X21           0.000000      0.000000
X22           1.000000      0.000000
X23           1.000000      0.000000

```

X24	0.000000	0.000000
X25	1.000000	0.000000
X31	0.000000	3.000000
X32	1.000000	0.000000
X33	1.000000	0.000000
X34	0.000000	2.000000
X35	1.000000	0.000000
X41	0.000000	2.000000
X42	0.000000	0.000000
X43	0.000000	0.000000
X44	2.000000	0.000000
X45	1.000000	0.000000
Row	Slack or Surplus	Dual Price
1	67.00000	1.000000
2	0.000000	6.000000
3	0.000000	3.000000
4	0.000000	2.000000
5	0.000000	1.000000
6	1.000000	0.000000
7	0.000000	3.000000
8	0.000000	0.000000
9	0.000000	4.000000
10	0.000000	1.000000
11	0.000000	0.000000
12	1.000000	0.000000
13	1.000000	0.000000
14	1.000000	0.000000
15	1.000000	0.000000
16	0.000000	4.000000
17	0.000000	1.000000
18	1.000000	0.000000
19	0.000000	3.000000
20	0.000000	7.000000
21	0.000000	0.000000
22	1.000000	0.000000
23	0.000000	2.000000
24	1.000000	0.000000
25	1.000000	0.000000
26	1.000000	0.000000
27	0.000000	5.000000
28	0.000000	0.000000
29	0.000000	0.000000

The dual prices on the first five constraints essentially provide us with the needed market clearing prices. To avoid ties, we may wish to add or subtract a small number to each of these prices. We claim that acceptable market clearing prices for objects 1, 2, 3, 4 and 5 are 5, 5, 3, 0, and 0, respectively.

Now note that, at these prices, the market clears. Bidder 1 is awarded the sole unit of object 1 at a price of \$5.00. If the price were lower, bidder 4 could claim the unit. If the price were more than 6, then bidder 1's surplus on object 1 would be less than $9 - 6 = 3$. Therefore, he would prefer object 5 instead. Where his surplus is $3 - 0 = 3$. If object 2's price were less than 4, then bidder 4 could claim the unit. If the price were greater than 5, then bidder 3 would prefer to give up his type-2 unit (with surplus $8 - 5 =$

3) and take a type-4 unit, which has a surplus of $3 - 0 = 3$. Similar arguments apply to objects 3, 4, and 5.

15.5 Multi-Product Pricing Problems

When a vendor sets prices, they should take into account the fact that a buyer will tend to purchase a product or, more generally, a bundle of products that gives the buyer the best deal. In economics terminology, the vendor should assume buyers will maximize their utility. A reasonable way of representing buyer behavior is to make the following assumptions:

1. Prospective buyers can be partitioned into market segments (e.g., college students, retired people, etc.). Segments can be defined sufficiently small, so individuals in the same segment have the same preferences.
2. Each buyer has a reservation price for each possible combination (or bundle) of products he or she might buy.
3. Each buyer will purchase that single bundle for which his reservation price minus his cost is maximized.

A smart vendor will set prices to maximize his profits, subject to customers maximizing their utility as described in (1-3).

The following is a general model that allows a number of features:

- a) some segments (e.g., students) may get a discount from the list price;
- b) there may be a customer segment specific cost of selling a product (e.g., because of a tax or intermediate dealer commission);
- c) the vendor incurs a fixed cost if he wishes to sell to a particular segment;
- d) the vendor incurs a fixed cost if he wishes to sell a particular product, regardless of whom it is sold to.

Analyses or models such as we are about to consider, where we take into account how customers choose products based on prices that vendors set, or which products vendors make available, are sometimes known as consumer choice models.

The model is applied to an example involving a vendor wishing to sell seven possible bundles to three different market segments: the home market, students, and the business market. The vendor has

decided to give a 10% discount to the student segment and incurs a 5% selling fee for products sold in the home market segment:

```

MODEL:
  !Product pricing (PRICPROD);
  !Producer chooses prices to maximize producer
  surplus;
  !Each customer chooses the one
  product/bundle that maximizes consumer surplus;
SETS:
  CUST:
    SIZE, ! Each cust/market has a size;
    DISC, ! Discount off list price willing to
           give to I;
    DISD, ! Discount given to dealer(who sells
           full price);
    FM, ! Fixed cost of developing market I;
    YM, ! = 1 if we develop market I, else 0;
    SRP; ! Consumer surplus achieved by customer
          I;
  BUNDLE:
    COST, ! Each product/bundle has a cost/unit to
           producer;
    FP, ! Fixed cost of developing product J;
    YP, ! = 1 if we develop product J, else 0;
    PRICE, ! List price of product J;
    PMAX; ! Max price that might be charged;
  CXB( CUST, BUNDLE): RP, ! Reservation
                        price of customer I for product J;
    EFP, ! Effective price I pays for J, = 0
          if not bought;
    X; ! = 1 if I buys J, else 0;
ENDSETS
DATA:
  ! The customer/market segments;
  CUST = HOME    STUD    BUS;
  ! Customer sizes;
  SIZE = 4000    3000    3000;
  ! Fixed market development costs;
  FM = 15000    12000    10000;
  ! Discount off list price to each customer, 0 <= DISC < 1;
  DISC = 0      .1      0;
  ! Discount/tax off list to each dealer, 0
  <= DISD < 1;
  DISD = .05    0      0;
  BUNDLE = B1    B2    B3    B12    B13    B23    B123;
  ! Reservation prices;
  RP = 400    50    200    450    650    250    700
      200    200    50    350    250    250    400
      500    100    100    550    600    260    600;
  ! Variable costs of each product bundle;
  COST = 100    20    30    120    130    50    150;
  ! Fixed product development costs;

```

```

      FP = 30000 40000 60000 10000 20000 8000 0;
ENDDATA
!-----;
! The seller wants to maximize the profit
  contribution;
[PROFIT] MAX =
@SUM( CXB( I, J):
  SIZE( I) * EFP( I, J)           ! Revenue;
  - COST( J)* SIZE( I) * X( I, J)
    ! Variable cost;
  - EFP( I, J) * SIZE( I) * DISD( I))
    ! Discount to dealers;
  - @SUM( BUNDLE: FP * YP)
    ! Product development cost;
  - @SUM( CUST: FM * YM);
    ! Market development cost;
! Each customer can buy at most 1 bundle;
@FOR( CUST( I):
  @SUM( BUNDLE( J) : X( I, J)) <= YM( I);
  @BIN( YM( I));
);
! Force development costs to be incurred
  if in market;
@FOR( CXB( I, J): X( I, J) <= YP( J);
  ! for product J;
! The X's are binary, yes/no, 1/0 variables;
  @BIN( X( I, J));
);
! Compute consumer surplus for customer I;
@FOR( CUST( I): SRP( I)
  = @SUM( BUNDLE( J): RP( I, J) * X( I, J)
  - EFP( I, J));
! Customer chooses maximum consumer surplus;
@FOR( BUNDLE( J):
  SRP( I) >= RP( I, J)
  - ( 1 - DISC( I)) * PRICE( J)
  );
);
! Force effective price to take on proper value;
@FOR( CXB( I, J):
! zero if I does not buy J;
  EFP( I, J) <= X( I, J) * RP( I, J);
! cannot be greater than price;
  EFP( I, J) <= ( 1 - DISC( I)) * PRICE( J);
! cannot be less than price if bought;
  EFP( I, J) >= ( 1 - DISC( I)) * PRICE( J)
    - ( 1 - X( I, J)) * PMAX( J);
);
! Compute upper bounds on prices;
@FOR( BUNDLE( J): PMAX( J)
  = @MAX( CUST( I): RP( I, J)/(1 - DISC( I)));
);
END

```

The solution, in part, is:

Global optimal solution found at step:		146
Objective value:		3895000.
Branch count:		0
Variable	Value	Reduced Cost
PRICE(B1)	500.0000	0.0000000
PRICE(B2)	222.2222	0.0000000
PRICE(B3)	200.0000	0.0000000
PRICE(B12)	550.0000	0.0000000
PRICE(B13)	650.0000	0.0000000
PRICE(B23)	277.7778	0.0000000
PRICE(B123)	700.0000	0.0000000
X(HOME, B123)	1.000000	-2060000.
X(STUD, B23)	1.000000	-592000.0
X(BUS, B12)	1.000000	-1280000.

In summary, the home segment buys product bundle *B123* at a price of \$700. The student segment buys product bundle *B23* at a list price of \$277.78, (i.e., a discounted price of \$250). The business segment buys product bundle *B12* at a price of \$550.

The prices of all other bundles can be set arbitrarily large. You can verify each customer is buying the product bundle giving the best deal:

Cust	Reservation price minus actual price		
	B12	B23	B123
Hom	$450 - 550 = -100$	$250 - 277.78 = -27.78$	$700 - 700 = 0$
Std	$350 - 9 \cdot 550 = -145$	$250 - 9 \cdot 277.78 = 0$	$400 - 9 \cdot 700 = -230$
Bus	$550 - 550 = 0$	$260 - 277.78 = -17.78$	$600 - 700 = -100$

The vendor makes a profit of \$3,895,000. In contrast, if no bundling is allowed, the vendor makes a profit of \$2,453,000.

There may be other equilibrium solutions. However, the above solution is one that maximizes the profits of the vendor. An equilibrium such as this, where one of the players is allowed to select the equilibrium most favorable to that player, is called a Stackelberg equilibrium.

An implementation issue that one should be concerned with when using bundle pricing is the emergence of third party brokers who will buy your bundle, split it, and sell the components for a profit. For our example, a broker might buy the full bundle *B123* for \$700, sell the *B1* component for \$490 to the Business market, sell the *B2* component for \$190 (after discount) to the student market, sell the *B3* component to the Home market for \$190, and make a profit of $490 + 190 + 190 - 700 = \170 . The consumers should be willing to buy these components from the broker because their consumer surplus is \$10, as compared to the zero consumer surplus when buying the bundles. This generally legal (re-)selling of different versions of the products to consumers in ways not intended by the seller is sometimes known as a "gray market", as compared to a black market where clearly illegal sales take place. Bundle pricing is a generalization of quantity discount pricing (e.g., "buy one, get the second one for half price") where the bundle happens to contain identical products. The same sort of gray market possibility exists with quantity discounts. The seller's major protection against gray markets is to make sure that the transaction costs of breaking up and reselling the components are too high. For example, if the only way of buying software is pre-installed on a computer, then the broker would have to setup an extensive operation to uninstall the bundled software and then reinstall the reconfigured software.

15.6 General Equilibrium Models of An Economy

When trade agreements are being negotiated between countries, each country is concerned with how the agreement will affect various industries in the country. A tool frequently used for answering such questions is the general equilibrium model. In a general equilibrium model of an economy, one wants to simultaneously determine prices and production quantities for several goods. The goods are consumed by several market sectors. Goods are produced by a collection of processes. Each process produces one or more goods and consumes one or more goods. At an equilibrium, a process will be used only if the value of the goods produced at least equals the cost of the goods required by the process.

When two or more countries are contemplating lowering trade barriers, they may want to look at general equilibrium models to get some estimates of how various industries will fare in the different countries as the markets open up.

An example based on two production processes producing four goods for consumption in four consumption sectors is shown below. Each sector has a demand curve for each good, based on the price of each good. Each production process in the model below is linear (i.e., it produces one or more goods from one or more of the other goods in a fixed proportion). A production process will not be used if the cost of raw materials and production exceeds the market value of the goods produced. The questions are: What is the clearing price for each good, and how much of each production process will be used?

MODEL:

```
! General Equilibrium Model of an economy, (GENEQLB1);
! Data based on Kehoe, Math Prog, Study 23(1985);
! Find clearing prices for commodities/goods and
  equilibrium production levels for processes in
  an economy;
```

SETS:

```
GOOD: PRICE, H;
SECTOR;
GXS( GOOD, SECTOR): ALPHA, W;
PROCESS: LEVEL, RC;
GXP( GOOD, PROCESS): MAKE;
```

ENDSETS

```

DATA:
  GOOD = 1..4; SECTOR = 1..4;
  ! Demand curve parameter for each good i & sector j;
  ALPHA =
    .5200 .8600 .5000 .0600
    .4000 .1 .2 .25
    .04 .02 .2975 .0025
    .04 .02 .0025 .6875;
  ! Initial wealth of good i by for sector j;
  W =
    50 0 0 0
    0 50 0 0
    0 0 400 0
    0 0 0 400;
  PROCESS= 1 2; ! There are two processes to make goods;
  !Amount produced of good i per unit of process j;
  MAKE =
    6 -1
    -1 3
    -4 -1
    -1 -1;
  ! Weights for price normalization constraint;
  H = .25 .25 .25 .25;
ENDDATA
!-----;
! Variables:
  LEVEL(p) = level or amount at which we operate
             process p.
  RC(p) = reduced cost of process p,
          = cost of inputs to process p - revenues from outputs
            of process p, per unit.
  PRICE(g) = equilibrium price for good g;
! Constraints;
! Supply = demand for each good g;
@FOR( GOOD( G):
  @SUM( SECTOR( M): W( G, M))
  + @SUM( PROCESS( P): MAKE( G, P) * LEVEL( P))
  = @SUM( SECTOR( S):
    ALPHA( G, S) *
    @SUM( GOOD( I): PRICE( I) * W( I, S))/ PRICE( G));
);
! Each process at best breaks even;
@FOR( PROCESS( P):
  RC(P) = @SUM( GOOD( G): - MAKE( G, P) * PRICE( G));
! Complementarity constraints. If process p
  does not break even(RC > 0), then do not use it;
  RC(P)*LEVEL(P) = 0;
);
! Prices scale to 1;
@SUM( GOOD( G): H( G) * PRICE( G)) = 1;
! Arbitrarily maximize some price to get a unique solution;
Max = PRICE(1);

```

END

The complementarity constraints, $RC(P) * LEVEL(P) = 0$, make this model difficult to solve for a traditional nonlinear solver. If the Global Solver option in LINGO is used, then this model is easily solved, giving the clearing prices:

PRICE (1)	1.100547
PRICE (2)	1.000000
PRICE (3)	1.234610
PRICE (4)	0.6648431

and the following production levels for the two processes:

LEVEL (1)	53.18016
LEVEL (2)	65.14806

This model in fact has three solutions, see Kehoe (1985). The other two are

PRICE (1)	0.6377
PRICE (2)	1.0000
PRICE (3)	0.1546
PRICE (4)	2.2077

and:

Variable	Value
PRICE (1)	1.0000
PRICE (2)	1.0000
PRICE (3)	1.0000
PRICE (4)	1.0000

Which solution you get may depend upon the objective function provided.

15.7 Transportation Equilibria

When designing a highway or street system, traffic engineers usually use models of some sophistication to predict the volume of traffic and the expected travel time on each link in the system. For each link, the engineers specify estimated average travel time as a nondecreasing function of traffic volume on the link.

The determination of the volume on each link is usually based upon a rule called Wardrop's Principle: If a set of commuters wish to travel from A to B , then the commuters will take the shortest route in the travel time sense. The effect of this is, if there are alternative routes from A to B , commuters will distribute themselves over these two routes, so either travel times are equal over the two alternates or none of the A to B commuters use the longer alternate.

As an example, consider the network in Figure 15.2. Six units of traffic (e.g., in thousands of cars) want to get from A to B .


This is a network with congestion, that is, travel time on a link increases as the volume of traffic increases. The travel time on any link as a function of the traffic volume is given in the following table:

For All Traffic Volumes Less-Than-or-Equal-To	Link Travel Time in Minutes				
	AB	AC	BC	BD	CD
2	20	52	12	52	20
3	30	53	13	53	30
4	40	54	14	54	40

The dramatically different functions for the various links might be due to such features as number of lanes or whether a link has traffic lights or stop signs.

We are interested in how traffic will distribute itself over the three possible routes ABD , ACD , and $ABCD$ if each unit behaves individually optimally. That is, we want to find the flows for which a user is indifferent between the three routes:

Figure 15.2 A Transportation Network



This can be formulated as an LP analogous to the previous equilibrium problems if the travel time schedules are interpreted as supply curves.

Define variables as follows. Two-letter variable names (e.g., AB or CD) denote the total flow along a given arc (e.g., the arc AB or the arc CD). Variables with a numeric suffix denote the incremental flow along a link. For example, $AB2$ measures flow up to 2 units on link $A \rightarrow B$. $AB3$ measures the incremental flow above 2, but less than 3.

The formulation is then:

```

MIN = 20 * AB2 + 30 * AB3 + 40 * AB4 + 52 * AC2
+ 53 * AC3 + 54 * AC4 + 12 * BC2 + 13 * BC3
+ 14 * BC4 + 52 * BD2 + 53 * BD3 + 54 * BD4
+ 20 * CD2 + 30 * CD3 + 40 * CD4;
! Minimize sum of congestion of incremental units;
- AB2 - AB3 - AB4 + AB = 0;
!Definition of AB;
- AC2 - AC3 - AC4 + AC = 0;
- BC2 - BC3 - BC4 + BC = 0;
- BD2 - BD3 - BD4 + BD = 0;
- CD2 - CD3 - CD4 + CD = 0;
AB + AC = 6;
!Flow out of A;
AB - BC - BD = 0;
!Flow through B;
AC + BC - CD = 0;
!Flow through C;
BD + CD = 6;
!Flow into D;
AB2 <= 2;
!Definition of the steps in;
AB3 <= 1;
!supply cost schedule;
AB4 <= 1;
AC2 <= 2;
AC3 <= 1;
AC4 <= 1;
BC2 <= 2;
BC3 <= 1;
BC4 <= 1;
BD2 <= 2;
BD3 <= 1;
BD4 <= 1;
CD2 <= 2;
CD3 <= 1;
CD4 <= 1;

```

The objective requires a little bit of explanation. It minimizes the incremental congestion seen by each incremental individual unit as it “selects” its route. It does not take into account the additional congestion that the incremental unit imposes on units already taking the route. Because additional traffic typically hurts rather than helps, this suggests this objective will understate true total congestion costs. Let us see if this is the case.

The solution is:

Objective Value	452.0000000	
Variable	Value	Reduced Cost
AB2	2.000000	0.000000
AB3	1.000000	0.000000
AB4	1.000000	0.000000
AC2	2.000000	0.000000
AC3	0.000000	1.000000
AC4	0.000000	2.000000
BC2	2.000000	0.000000
BC3	0.000000	1.000000
BC4	0.000000	2.000000
BD2	2.000000	0.000000
BD3	0.000000	1.000000
BD4	0.000000	2.000000
CD2	2.000000	0.000000
CD3	1.000000	0.000000
CD4	1.000000	0.000000
AB	4.000000	0.000000
AC	2.000000	0.000000
BC	2.000000	0.000000
BD	2.000000	0.000000
CD	4.000000	0.000000
Row	Slack	Dual Prices
2)	0.000000	40.000000
3)	0.000000	52.000000
4)	0.000000	12.000000
5)	0.000000	52.000000
6)	0.000000	40.000000
7)	0.000000	-92.000000
8)	0.000000	52.000000
9)	0.000000	40.000000
10)	0.000000	0.000000
11)	0.000000	20.000000
12)	0.000000	10.000000
13)	0.000000	0.000000
14)	0.000000	0.000000
15)	1.000000	0.000000
16)	1.000000	0.000000
17)	0.000000	0.000000
18)	1.000000	0.000000
19)	1.000000	0.000000
20)	0.000000	0.000000
21)	1.000000	0.000000
22)	1.000000	0.000000
23)	0.000000	20.000000
24)	0.000000	10.000000
25)	0.000000	0.000000

Notice 2 units of traffic take each of the three possible routes: *ABD*, *ABCD*, and *ACD*. The travel time on each route is 92 minutes. This agrees with our understanding of an equilibrium (i.e., no user is motivated to take a different route). The total congestion is $6 \times 92 = 552$, which is greater than the 452 value of the

objective of the LP. This is, as we suspected, because the objective measures the congestion incurred by the incremental unit. The objective function value has no immediate practical interpretation for this formulation. In this case, the objective function is simply a device to cause Wardrop's principle to hold when the objective is optimized.

The solution approach based on formulating the traffic equilibrium problem as a standard LP was presented mainly for pedagogical reasons. For larger, real-world problems, there are highly specialized procedures (cf., Florian (1977)).

15.7.1 User Equilibrium vs. Social Optimum

We shall see, for this problem, the solution just displayed does not minimize total travel time. This is a general result: the so-called user equilibrium, wherein each player in a system behaves optimally, need not result in a solution as good as a social optimum, which is best overall in some sense. Indeed, the user equilibrium need not even be Pareto optimal. In order to minimize total travel time, it is useful to prepare a table of total travel time incurred by users of a link as a function of link volume. This is done in the following table, where "Total" is the product of link volume and travel time at that volume:

Total and Incremental Travel Time Incurred on a Link										
Traffic Volume	AB		AC		BC		BD		CD	
	Total	Rate/Unit	Total	Rate/Unit	Total	Rate/Unit	Total	Rate/Unit	Total	Rate/Unit
2	40	20	104	52	24	12	104	52	40	20
3	90	50	159	55	39	15	159	55	90	50
4	160	70	216	57	56	17	216	57	160	70

The appropriate formulation is:

$$\begin{aligned}
 \text{MIN} &= 20 * AB2 + 50 * AB3 + 70 * AB4 + 52 * AC2 \\
 &+ 55 * AC3 + 57 * AC4 + 12 * BC2 + 15 * BC3 \\
 &+ 17 * BC4 + 52 * BD2 + 55 * BD3 + 57 * BD4 \\
 &+ 20 * CD2 + 50 * CD3 + 70 * CD4; \\
 &\text{! Minimize total congestion;} \\
 &- AB2 - AB3 - AB4 + AB = 0 ; \\
 &\text{! Definition of AB;} \\
 &- AC2 - AC3 - AC4 + AC = 0 ; \\
 &\text{! and AC;} \\
 &BC2 - BC3 - BC4 + BC = 0 ; \\
 &\text{! BC;} \\
 &- BD2 - BD3 - BD4 + BD = 0 ; \\
 &\text{! BD;} \\
 &- CD2 - CD3 - CD4 + CD = 0 ; \\
 &\text{! and CD;} \\
 &AB + AC = 6; \\
 &\text{! Flow out of A;} \\
 &AB - BC - BD = 0; \\
 &\text{! Flow through B;} \\
 &AC + BC - CD = 0 ; \\
 &\text{! Flow through C;} \\
 &BD + CD = 6 ;
 \end{aligned}$$

```

! Flow into D;
  AB2 <= 2;
! Steps in supply schedule;
  AB3 <= 1;
  AB4 <= 1;
  AC2 <= 2;
  AC3 <= 1;
  AC4 <= 1;
  BC2 <= 2;
  BC3 <= 1;
  BC4 <= 1;
  BD2 <= 2;
  BD3 <= 1;
  BD4 <= 1;
  CD2 <= 2;
  CD3 <= 1;
  CD4 <= 1;

```

The solution is:

```

Optimal solution found at step:      16
Objective value:                    498.0000

```

Variable	Value	Reduced Cost
AB2	2.000000	0.000000
AB3	1.000000	0.000000
AB4	0.000000	0.000000
AC2	2.000000	0.000000
AC3	1.000000	0.000000
AC4	0.000000	0.000000
BC2	0.000000	0.000000
BC3	0.000000	27.00000
BC4	0.000000	29.00000
BD2	2.000000	0.000000
BD3	1.000000	0.000000
BD4	0.000000	0.000000
CD2	2.000000	0.000000
CD3	1.000000	0.000000
CD4	0.000000	0.000000
AB	3.000000	0.000000
AC	3.000000	0.000000
BC	0.000000	1.000000
BD	3.000000	0.000000
CD	3.000000	0.000000

Row	Slack or Surplus	Dual Price
1	498.0000	1.000000
2	0.000000	70.00000
3	0.000000	57.00000
4	0.000000	-12.00000
5	0.000000	57.00000
6	0.000000	70.00000
7	0.000000	-70.00000
8	0.000000	0.000000
9	0.000000	13.00000

10	0.0000000	-57.00000
11	0.0000000	50.00000
12	0.0000000	20.00000
13	1.0000000	0.0000000
14	0.0000000	5.0000000
15	0.0000000	2.0000000
16	1.0000000	0.0000000
17	2.0000000	0.0000000
18	1.0000000	0.0000000
19	1.0000000	0.0000000
20	0.0000000	5.0000000
21	0.0000000	2.0000000
22	1.0000000	0.0000000
23	0.0000000	50.00000
24	0.0000000	20.00000
25	1.0000000	0.0000000

An interesting feature is no traffic uses link BC . Three units each take routes ABD and ACD . Even more interesting is the fact that the travel time on both routes is 83 minutes. This is noticeably less than the 92 minutes for the previous solution. With this formulation, the objective function measures the total travel time incurred. Note $498/6 = 83$.

If link BC were removed, this latest solution would also be a user equilibrium because no user would be motivated to switch routes. The interesting paradox is that, by adding additional capacity, in this case link BC , to a transportation network, the total delay may actually increase. This is known as Braess's Paradox (cf., Braess (1968) or Murchland (1970)). Murchland claims that this paradox was observed in Stuttgart, Germany when major improvements were made in the road network of the city center. When a certain cross street was closed, traffic got better.

To see why the paradox occurs, consider what happens when link BC is added. One of the 3 units taking route ABD notices that travel time on link BC is 12 and time on link CD is 30. This total of 42 minutes is better than the 53 minutes the unit is suffering in link BD , so the unit replaces link BD in its route by the sequence BCD . At this point, one of the units taking link AC observes it can reduce its delay in getting to C by replacing link AC (delay 53 minutes) with the two links AB and BC (delay of $30 + 12 = 42$). Unfortunately (and this is the cause of Braess's paradox), neither of the units that switched took into account the effect of their actions on the rest of the population. The switches increased the load on links AB and CD , two links for which increased volume dramatically increases the travel time of everyone. The general result is, *when individuals each maximize their own objective function, the obvious overall objective function is not necessarily maximized*. Braess Paradox is a variation of the Prisoner's Dilemma. If the travelers "cooperate" with each other and avoid link BC , then all travelers would be better off.

15.8 Equilibria in Networks as Optimization Problems

For physical systems, it is frequently the case that the equilibrium state is one that minimizes the energy loss or the energy level. This is illustrated in the model below for an electrical network. Given a set of resistances in a network, if we minimize the energy dissipated, then we get the equilibrium flow. In the network model corresponding to this model, a voltage of 120 volts is applied to node 1. The dual prices at a node are the voltages at that node:

```

MODEL:
! Model of voltages and currents in a Wheatstone
  Bridge;
DATA:
R12 = 10;
R13 = 15;
R23 = 8;
R32 = 8;
R24 = 20;
R34 = 16;
ENDDATA
! Minimize the energy dissipated;
MIN = (I12 * I12 * R12 + I13 * I13 * R13
      + I23 * I23 * R23 + I24 * I24 * R24
      + I32 * I32 * R32 + I34 * I34 * R34) / 2
      - 120 * I01;
[NODE1] I01 = I12 + I13;
[NODE2] I12 + I32 = I23 + I24;
[NODE3] I13 + I23 = I32 + I34;
[NODE4] I24 + I34 = I45;
END

```

```

Optimal solution found at step:          13
Objective value:                       -479.5393

```

Variable	Value	Reduced Cost
R12	10.00000	0.0000000
R13	15.00000	0.0000000
R23	8.000000	0.0000000
R32	8.000000	0.0000000
R24	20.00000	0.0000000
R34	16.00000	0.0000000
I12	4.537428	0.0000000
I13	3.454894	0.0000000
I23	0.8061420	0.1504372E-05
I24	3.731286	0.2541348E-05
I32	0.0000000	6.449135
I34	4.261036	0.1412317E-05
I01	7.992322	0.0000000
I45	7.992322	0.0000000

Row	Slack or Surplus	Dual Price
1	-479.5393	1.000000
NODE1	0.0000000	120.0000
NODE2	0.0000000	74.62572
NODE3	0.0000000	68.17658
NODE4	0.0000000	0.0000000

15.8.1 Equilibrium Network Flows

Another network setting involving nonlinearities is in computing equilibrium flows in a network. Hansen, Madsen, and H.B. Nielsen (1991) give a good introduction. The laws governing the flow depend upon the type of material flowing in the network (e.g., water, gas, or electricity). Equilibrium in a network is described by two sets of values:

- a) flow through each arc;
- b) pressure at each node (e.g., voltage in an electrical network).

At an equilibrium, the values in (a) and (b) must satisfy the rules or laws that determine an equilibrium in a network. In general terms, these laws are:

- i. for each node, standard conservation of flow constraints apply to the flow values;
- ii. for each arc, the pressure difference between its two endpoint nodes is related to the flow over the arc and the resistance of the arc.

In an electrical network, for example, condition (ii) says the voltage difference, V , between two points connected by a wire with resistance in ohms, R , over which a current of I amperes flows, must satisfy the constraint: $V = I \times R$.

The constraints (ii) tend to be nonlinear. The following model illustrates by computing the equilibrium in a simple water distribution network for a city. Pumps apply a specified pressure at two nodes, G and H . At the other nodes, water is removed at specified rates. We want to determine the implied flow rate on each arc and the pressure at each node:

```

MODEL:
! Network equilibrium NETEQL2:based on
  Hansen et al., Mathematical Programming, vol. 52, no.1;
SETS:
NODE: DL, DU, PL, PU, P, DELIVER; ! P = Pressure at this node;
ARC( NODE, NODE): R, FLO; ! FLO = Flow on this arc;
ENDSETS
DATA:
NODE =    A,    B,    C,    D,    E,    F,    G,    H;
! Lower & upper limits on demand at each node;
DL =     1     2     4     6     8     7 -9999 -9999;
DU =     1     2     4     6     8     7  9999  9999;
! Lower & upper limits on pressure at each node;
PL =     0     0     0     0     0     0  240  240;
PU =  9999  9999  9999  9999  9999  9999  240  240;

! The arcs available and their resistance parameter;
ARC = B A, C A, C B, D C, E D, F D, G D, F E, H E, G F, H F;
R =   1,   25,   1,   3,  18,  45,   1,  12,   1,  30,   1;

PPAM = 1; ! Compressibility parameter;
!For incompressible fluids and electricity: PPAM = 1, for gases: PPAM
= 2;
FPAM = 1.852; !Resistance due to flow parameter;
!      electrical networks:  FPAM = 1;
!      other fluids:  1.8 <= FPAM <= 2;
! For optimization networks: FPAM=0, for arcs with flow>=0;
ENDDATA

```

```

@FOR( NODE( K): ! For each node K;
  ! Bound the pressure;
  @BND( PL(K), P(K), PU(K));
  ! Flow in = amount delivered + flow out;
  @SUM( ARC( I, K): FLO( I, K)) = DELIVER( K) +
  @SUM( ARC( K, J): FLO( K, J));
  ! Bound on amount delivered at each node;
  @BND( DL(K), DELIVER(K), DU(K));
);

@FOR( ARC( I, J):
  ! Flow can go either way;
  @FREE( FLO(I,J));
  ! Relate pressures at 2 ends to flow over arc;
  P(I)^ PPAM - P(J)^ FPAM =
  R(I,J)* @SIGN(FLO(I,J))* @ABS( FLO(I,J))^ FPAM;);
END

```

Verify the following solution satisfies conservation of flow at each node and the pressure drop over each arc satisfies the resistance equations of the model:

Variable	Value	
Feasible solution found at step:		22
PPAM	1.000000	
FPAM	1.852000	
P(A)	42.29544	
P(B)	42.61468	
P(C)	48.23412	
P(D)	158.4497	
P(E)	188.0738	
P(F)	197.3609	
P(G)	240.0000	
P(H)	240.0000	
FLO(B, A)	0.5398153	
FLO(C, A)	0.4601847	
FLO(C, B)	2.539815	
FLO(D, C)	7.000000	
FLO(E, D)	1.308675	
FLO(F, D)	0.9245077	
FLO(F, E)	0.8707683	
FLO(G, D)	10.76682	
FLO(G, F)	1.209051	
FLO(H, E)	8.437907	
FLO(H, F)	7.586225	

15.9 Problems


- Producer B in the two-producer, two-consumer market at the beginning of the chapter is actually a foreign producer. The government of the importing country is contemplating putting a \$0.60 per unit tax on units from Producer B .
 - How is the formulation changed?
 - How is the equilibrium solution changed?
- An organization is interested in selling five parcels of land, denoted A , B , C , D , and E , which it owns. It is willing to accept offers for subsets of the five parcels. Three buyers, x , y , and z are interested in making offers. In the privacy of their respective offices, each buyer has identified the maximum price he would be willing to pay for various combinations. This information is summarized below:

Buyer	Parcel Combination	Maximum Price
x	A, B, D	95
x	C, D, E	80
y	B, E	60
y	A, D	82
z	B, D, E	90
z	C, E	71

Each buyer wants to buy at most one parcel combination. Suppose the organization is a government and would like to maximize social welfare. What is a possible formulation based on an LP for holding this auction?

- Commuters wish to travel from points A , B , and C to point D in the network shown in Figure 15.3:

Figure 15.3 A Travel Network



Three units wish to travel from A to D , two units from B to D , and one from C to D . The travel times on the five links as a function of volume are:

For All Volumes Less-Than-or-Equal-To:	Link Travel Time in Minutes				
	AC	AD	BC	BD	CD
2	21	50	17	40	12
3	31	51	27	41	13
4	41	52	37	42	14

- a) Display the LP formulation corresponding to a Wardrop's Principle user equilibrium.
 - b) Display the LP formulation useful for the total travel time minimizing solution.
 - c) What are the solutions to (a) and (b)?
4. In the sale of real estate and in the sale of rights to portions of the radio frequency spectrum, the value of one item to a buyer may depend upon which other items the buyer is able to buy. A method called a combinatorial auction is sometimes used in such cases. In such an auction, a bidder is allowed to submit a bid on a combination of items. The seller is then faced with the decision of which combination of these "combination" bids to select. Consider the following situation. The Duxbury Ranch is being sold for potential urban development. The ranch has been divided into four parcels, A , B , C , and D for sale. Parcels A and B both face major roads. Parcel C is a corner parcel at the intersection of the two roads. D is an interior parcel with a narrow access to one of the roads. The following bids have been received for various combinations of parcels:

Bid No.	Amount	Parcels Desired
1	\$380,000	A, C
2	\$350,000	A, D
3	\$800,000	A, B, C, D
4	\$140,000	B
5	\$120,000	B, C
6	\$105,000	B, D
7	\$210,000	C
8	\$390,000	A, B
9	\$205,000	D
10	\$160,000	A

Which combination of bids should be selected to maximize revenues, subject to not selling any parcel more than once?

5. Perhaps the greatest German writer ever was Johann Wolfgang von Goethe. While trying to sell one of his manuscripts to a publisher, Vieweg, he wrote the following note to the publisher: "Concerning the royalty, we will proceed as follows: I will hand over to Mr. Counsel Bottiger a sealed note, which contains my demand, and I wait for what Mr. Vieweg will suggest to offer for my work. If his offer is lower than my demand, then I take my note back, unopened, and the negotiation is broken. If, however, his offer is higher, then I will not ask for more than what is written in the note to be opened by Mr. Bottiger."(see Moldovanu and Tietzel (1998)). If you were the publisher, how would you decide how much to bid?

